Temperature Control Programs to Select the Optimal Control Input for a Precision Water Bath

Junko Kawahito and Shuko Fujieda*

The Doctoral Research Course in Human Culture,
Ochanomizu University, Bunkyo-ku, Tokyo
* Department of Chemistry, Faculty of Science,
Ochanomizu University, Bunkyo-ku, Tokyo

(Received September 5, 1991)

Abstract

Programs to control the temperature of a precision water bath for versatile uses are reported in which optimal control inputs were selected. Some popular instruments and hand-made interface circuits were used. The hardware was simplified as possible, and the temperature of the bath water was controlled to be kept within a narrow range of fluctuation in the software. The heat exchange of the bath with the surroundings, non-stationary heat flow from warm stirrer motors, and other thermal factors were considered to be a part of heat sources. Cooling water a little lower than the set temperature was circulated in the water bath. The difference between the observed temperature and the set temperature was acquired to a small microcomputer every second via a thermistor used as a temperature sensor. The thermal behavior of the bath water was simulated by a Newtonian cooling equation. From several data points, the temperature to be acquired at the next point was calculated by the method of least squares, and the control input was given to the heater element after power amplification. The programs were written in BASIC and amounted to about 5 kbyte. The temperature was controlled within $\pm 3\sim 4\times 10^{-4}$K at 25°C and 37°C, and the range was almost the same as that previously reported, or narrower.

Introduction

Scientific experiments are often carried out at constant temperature. However, a commercially available water bath which is kept within ±0.001 K of fluctuation range may be expensive. It is easy to construct simply a temperature control system for a precision water bath, provided that a small microcomputer and common instruments are used.

In the heat exchange calorimetry,[1] originated by Nakanishi and Fujieda, sample and reference vessels are fixed differentially in a water bath, and the heat evolved in the sample vessel is exchanged freely with the water. The thermal fluctuation of the water should be kept within a narrow range, because of the operation of the calorimeter in non-air-conditioned laboratories. Analog differentiation for estimating the rate of heat evolution is usually sensitive to minor fluctuations.

The modified PID (proportional, integral, and derivative, respectively) control system was developed to apply to the heat exchange calorimeter.[2] Thermal constancy was practically maintained in spite of thermal disturbances. However, all the P, I, and D constants were adjusted empirically by such experimental conditions as the volume and the set temperature of the bath water, etc., and classified in each case.

On the other hand, a general control system which was not affected by the empirical factors was proposed.[3] Cooling water a little lower than the set temperature was continuously circulated in the water bath. The deviation of the observed temperature of the bath water to the set temperature was fed into a small microcomputer every second. The process of the temperature deviations was simulated by the Newtonian equation with the same concept as the heat exchange calorimetry. The optimal control input to the water bath was calculated and given to the heater circuit, so that the deviation might be minimum or zero. The hardware was simplified as possible and consisted of commonly available instruments and simple handmade circuits.

In the present report, the control programs, by which the sophisticated control of the bath water was carried out, were explained in details. Though special conditions were not required, the results of the test runs in the practical use were also shown satisfactorily.

General Considerations

The fundamental concept of the heat exchange calorimetry was applied to express thermal behavior of the bath water, which may be considered to be a large vessel allowed to exchange heat freely with the surroundings. The thermal phenomena in the bath are expressed by the following differential equation

$$\frac{\mathrm{d}T_{\mathrm{bath}}}{\mathrm{d}t}=\frac{q}{W}-\alpha T_{\mathrm{bath}} \tag{1}$$

where T_{bath} is the temperature of the bath water, t the time, q the rate of heat evolution, W the effective heat capacity, and α a constant which gives the extent of heat exchange with the surroundings and is always positive. In the present report, W, instead of C, was used to express the effective heat capacity for preventing from confusion with a constant in eqn. (3). As q/W is not a constant in a chemical reaction, eqn. (1) cannot be solved mathematically. In the heat exchange calorimetry, it was treated by analog computations without solving mathematically. However, in a short time range and small temperature change, q/W was assumed to be constant to solve eqn. (1) mathematically. Then

$$T_{\text{bath}}=\frac{q}{\alpha W}(1-\exp(-\alpha t)) \tag{2}$$

From eqn. (2) and the temperature deviation, T_{dev}, observed as difference of T_{bath} and T_{set}, where T_{set} is the set temperature, the following eqn. (3) may be obtained as the general expression,

$$y=A\exp(Bt)+C \tag{3}$$

where $y=T_{\text{dev}}$, $A=q/\alpha W$, $B=-\alpha$, $C=T_{\text{set}}-q/\alpha W$. B is always negative, because α is positive from the physical meaning. Under the condition, y asymptotically approaches C at large t, regardless of the sign of A. When T_{bath} is equal to T_{set}, T_{dev} may be zero. Therefore, if $C=0$,

$$y=A\exp(Bt) \tag{4}$$

and eqn. (5) is obtained from the logarithmic calculation of eqn. (4):

$$\ln y=\ln A+Bt \tag{5}$$

The constants of A and B are calculated, as usual, by using the method of least squares from several observed data sets, (t_x, y_x) $(x=0, 1, \cdots, n)$. Then y_{n+1}, the deviation to be expected at t_{n+1}, is obtained. A control input on the basis of the calculated y_{n+1} was given to the bath water via the heater circuit.

Experimental

apparatus

The temperature control system assembled in this work is shown in Fig. 1.

A glass box $23\times30\times27\ \text{cm}^3$ in size was used as the water bath, whose all the six surfaces were covered with 5 cm thick Styrofoam insulator boards. A thermistor for measuring T_{bath}, screws of a motor-driven stirrer for agitation of the bath water, a glass spiral tubing for the cooling water, and a heater element were included in the bath.

The thermistor (NLB, Shibaura Denshi Co., Tokyo) comprised one arm of a Wheatstone bridge. The unbalanced voltage corresponding to T_{dev} was adjusted to the maximum of ±2 V by a preamplifier and fed into a small microcomputer (M5, SORD Co., Tokyo, operated with Z80A, Zilog) every second. Temperature change of 1.5×10^{-5}K in the bath water corresponded to 1 digit of input signal in the case of ±0.3 mV full scale of the preamplifier.



Fig. 1. Schematic diagram of the control system. Th; thermistor, H; heater, C; cooler, ADC; analog-to-digital converter, PIO; programmable parallel input-output, DAC; digital-to-analog converter.

The cooling water kept 2°C lower than the given T_{set} was circulated in the water bath at a flow rate of 1.5 dm^{3}min^{-1}. Heat leakage and heat input from warmed-up motors of the agitator or magnetic stirrers to the water bath were considered to be a part of cooling or heating. Only the difference of T_{bath} and T_{set} was observed in the presented system. Optimal signal inputs to the bath were calculated from the acquired data by using the proposed expression and given to the heater element in the bath via a power amplifier circuit.

Software

All programs were written in BASIC and amounted to ca. 5 kbyte. A simplified flow chart of the programs is shown in Fig. 2. The number of data sets for the calculation of numerical fitting of T_{dev} was examined,[3] four being selected as the recommended number in this report. Variance of the T_{dev} was approximated with eqn. (5), and the value of y_{n+1} was calculated. The optimal control input was calculated from y_{n+1} and/or the coefficients obtained in eqn. (5), and given to the heater circuit. Each variable was displayed on the monitor.

Detailed flow charts of two subroutines, Sub. (1) and Sub. (2), are shown in Figs. 3 and 4, in which ∗1～∗9 mean detailed explanation.

Sub. (1): The fitting of the temperature deviations to eqn. (5).

∗1 If the T_{dev} acquired into the microcomputer, written as *TEMP* in programs, was negative, the logarithm of $|TEMP|$ was calculated. If $TEMP=0$, then $\ln(TEMP)$ was treated as zero for convenience.

∗2 The coefficients in eqn. (5), A and B, were calculated by the method of least squares. In eqn. (3), B was considered to be always nega-



Fig. 2. A simplified flow chart of control programs.



Fig. 3. A flow chart for fitting of the temperature deviations to eqn. (5).

tive. However, B was positive in practice, when T_{bath} moved to part from T_{set}.

*3 The adequate sign of A was determined with the latest *TEMP* and/or A calculated at one second before.

*4 The y_{n+1} or the expected T_{dev}, written as *YEXP* in Figs. 3 and 4, was calculated by eqn. (5) from the obtained A and B.

Sub. (2): The calculation of optimal inputs to the heater circuit.

The parameter B reflected the change of *TEMP*'s; that is, its absolute value showed the rate of change, and its sign indicated the direction of the change against T_{set}. *BB* is defined as the value obtained at one second before. B and *BB* were used for the calculation of the optimal input to the heater, *HEAT%*, in order to correct speedily the rapid temperature change or the drift from T_{set}.

*5 The basic value of *HEAT%* was calculated from *YEXP* with a linear



Fig. 4. A flow chart for the calculation of optimal inputs to the heater.

relation.

*6 When $B \geqq 0$ and $YEXP \geqq 0$, i.e. the T_{bath} continued to be lower than T_{set}, *HEAT%* was doubled.

*7 By using *B*, *HEAT%* was modified. If *TEMP* rapidly approached zero, *HEAT%* was decreased. In the case that *TEMP* fell away from zero, either *A* or $A/2$ was added to *HEAT%*.

*8 If *TEMP* was close to zero and kept on approaching it, *HEAT%* which was already small was decreased further, in *BB* equal to *B* or larger than *B*.

*9 Three cases were considered as for signs of *B* and *BB*, and then *HEAT%* was calculated. Case (I): *TEMP* kept on falling away from zero. Case (II): *TEMP* kept the offset value. Case (III): the approach to zero was very slow.

Procedure

The bath was filled with the water of the temperature near the T_{set}.

The sensitivity of the preamplifier was selected corresponding to the required constancy of the T_{bath} to be controlled. Circulation of the cooling water, agitation of motor-driven stirrers, and working of the heater circuit were started. Then the temperature control system by the microcomputer began to run.

Results and Discussion

Test runs of the programs

The programs were tested with three types of water bath. A typical temperature trace controlled in the water bath by the proposed method is shown in Fig. 5. The bath was filled with 12 dm^3 of water and T_{set} was 25.0°C. The T_{bath} approached the T_{set} immediately. The fluctuation range was within $\pm 3 \sim 4 \times 10^{-4}$K. Thermal disturbances were given to the controlled system by adding hot or cold water, as shown by arrows (↑, ↓) in Fig. 5, and good recoveries were obtained. The response speed to the addition of hot water (↓) may depend on the power of the cooler.

Biological measurements may be often required at 37°C. Therefore, temperature of the bath water was also controlled at 37°C of T_{set} by using the proposed programs, and the result was similarly satisfied.

The second type of water bath, which was larger than the first, was tested under different conditions. It was filled with 28 dm^3 of water and covered with 3 cm thick Styrofoam insulator boards. Two motor-driven stirrers were used for the sufficient agitation. The third type of the water bath made of a plastic box without any cover boards was filled with 10 dm^3 of water and controlled at 25°C of T_{set} in the non-air-conditioned laboratory at a room temperature of 29±0.1°C. The T_{bath} undulated around the T_{set} owing to the influence of the surroundings, but the fluctuation range was within less than about $\pm 1 \times 10^{-3}$K.

The temperature control programs presented in this report provided a precision water bath under the conditions that the bath was simply covered



Fig. 5. The typical trace of the controlled temperature of the bath water.

and the water adequately agitated. The fluctuation was also small enough to utilize in a heat exchange calorimeter even for the estimation of small heat effect of ± 0.1 J or less.

The proposed system composed of simple and popular hardware may be widely used. The program may be more useful if it were stored in ROM.

References

[1] M. Nakanishi and S. Fujieda: *Anal. Chem.*, **44**, 574 (1972).
[2] S. Fujieda, M. Nakanishi and J. Kawahito: *Thermochim. Acta*, **157**, 183 (1990).
[3] S. Fujieda and J. Kawahito: *Thermochim. Acta*, **190**, 175 (1991).

List of the programs

```
100 ' *******************************************************
110 ' **                                                   **
120 ' **  TEMPERATURE CONTROL PROGRAM TO  SELECT THE        **
130 ' **  OPTIMAL CONTROL INPUT FOR A PRECISION WATER BATH  **
140 ' **                                                   **
150 ' **       by  J. Kawahito  and  S. Fujieda            **
160 ' **       Ochanomizu Univ.  (1991)                    **
170 ' **                                                   **
180 ' *******************************************************
190 '
200 ' **************************************
210 ' *       HARDWARE   INITIALIZING      *
220 ' **************************************
230 '
240 ' --- PORT B : to Acquire Temperature Deviations from the Water Bath ---
250       out &73,&9B
260 ' --- PORT A : to Send Control Signals to the Heater ---
270       out &63,&80
280       out &60,255  :  out &61,255
290 '
300 ' **************************************
310 ' *       INITIAL     DISPLAY          *
320 ' **************************************
330 '
340 cls : print "■"
350 for I=1 to 37
360     locate I,0   :  print "*"
370     locate I,22  :  print "*"
380 next I
390 for I=1 to 21
400     locate 1,I   :  print "*"
410     locate 37,I  :  print "*"
420 next I
430 '
440     locate 3,3     :   print "Data Aquisition via an ADC"
450     locate 10,4    :   print " ... +- 1999 full scale"
460     locate 3,7     :   print "Output of Control Signal to a DAC"
470     locate 10,8    :   print " ... 12 bit ; 0 to 10V"
480     locate 3,11    :   print "Data Numbers for the LSM Calculation"
```

```
490     locate 10,12   :   print " ... 4 "
500     locate 5,21    :   print " if END, please hit [E] key!"
510  '
520     locate 3,15    :   print "Time Interval of Data Sampling (s)"
530     locate 10,16   :   print " ... ";
540  input S  :  if S<1 then S=1   :  ' --- S is Sampling Interval ---
550     locate 14,16   :  print S;"sec"
560     locate 5,19    :  print "Ok?(y/n)"
570  SETTING$=inkey$ : if SETTING$="" then goto 570
580  if SETTING$="y" or SETTING$="Y" then goto 590
                 else if SETTING$="n" or SETTING$="N" then goto 540 else goto 570
590  cls
600  '
610  ' --- Setting of the Dimension for the Calculation of Least Squares ---
620    dim LN(3)
630    gosub $SAMPLING
640    for I=0 to 2    :    LN(I)=LTEMP    :    next I
650  '
660  ' ***************************************
670  ' *        MAIN    PROGRAM             *
680  ' ***************************************
690  '
700  ' --- Sampling Interval ---
710  S=S*60  :  event S
720  on event gosub 750
730      event on
740  goto 720
750  '
760     gosub $SAMPLING
770     gosub $LSM
780     gosub $CALC
790     gosub $HEATER
800  '
810     for I=0 to 2
820         LN(I)=LN(I+1)
830     next I
840  '
850  return
860  '
870  ' ***************************************
880  ' *  SUBROUTINES  FOR  INPUT / OUTPUT   *
890  ' ***************************************
900  '
910  $SAMPLING
```

```
920  ' *****  Acquisition of the Temperature Deviation  *****
930      '
940      INA=inp(&70) : INB=inp(&71)
950       TEMP0$=mid$(hex$(INA),3,1)
960       if INA<128 then TEMP1=1 else TEMP1=0      :      ' --- 1st digit ---
970       TEMP2=val(right$(hex$(INA),1))             :      ' --- 2nd digit ---
980       TEMP3=val(mid$(hex$(INB),3,1))             :      ' --- 3rd digit ---
990       TEMP4=val(right$(hex$(INB),1))             :      ' --- 4th digit ---
1000     ' --- [TEMP] is the Temperature Deviation ---
1010      TEMP=TEMP1*1000+TEMP2*100+TEMP3*10+TEMP4
1020     '
1030     ' --- What's [TEMP]'s Sign? ---
1040      if TEMP0$="0" or TEMP0$="A" or TEMP0$="B" then TEMP=(-1)*TEMP else TEMP=TEMP
1050     '
1060     ' --- Calculation of Natural Logarithm of [TEMP] ---
1070      if TEMP<>0 then LTEMP=ln(abs(TEMP)) else LTEMP=0
1080     '
1090 return
1100 '
1120 $HEATER
1130 ' ***** Output [HEAT%] to the Heater in the Water Bath *****
1140     '
1150      VOUT%=cint(4095-(4095*(HEAT%/1000)))
1160            VOUTA1%=val("&"+mid$(hex$(VOUT%),2,1))
1170            VOUTA2%=val("&"+mid$(hex$(VOUT%),3,1))
1180        VOUTA%=VOUTA1%*16+VOUTA2%
1190        VOUTB%=val("&"+right$(hex$(VOUT%),1))
1200     ' --- Send! ---
1210      out &60,VOUTA%   :   out &61,VOUTB%
1220      print tab(30);HEAT%
1230     '
1240 return
1250 '
1260 ' ****************************************
1270 ' *       SUBROUTINES    (1) & (2)       *
1280 ' ****************************************
1290 '
1300 $LSM
1310 ' *****  Numerical Fitting of Four [TEMP]s with the Least Squares  *****
1320     '
1330      LN(3)=LTEMP
1340      X1=0   :   X2=0   :   Y1=0   :   Y2=0
1350      for I=0 TO 3
1360           X1=X1+I
```

```
1370          X2=X2+I^2
1380          Y1=Y1+LN(I)
1390          Y2=Y2+LN(I)*I
1400     next I
1410     '
1420     '--- Calculations of [A] and [B] ---
1430     B=(4*Y2-X1*Y1)/(4*X2-X1^2)
1440     if TEMP<>0 THEN A=sgn(TEMP)*exp((Y1-B*X1)/4) else A=sgn(A)*exp((Y1-B*X1)/4)
1450     '
1460     '--- Calculation of [YEXP] as the Predicted Temperature at Next Sampling Time ---
1470     YEXP=A*exp(B*4*S)
1480        if YEXP>1999 then YEXP=1999
1490        if YEXP<-1999 then YEXP=-1999
1500     print "T=";TEMP;tab(9);"Y=";cint(YEXP);tab(17);
         "B=";cint(B*1000)/1000;tab(27);
1510     '
1520 return
1530 '
1540 $CALC
1550 ' *****  Calculation of the Optimal Input (as [HEAT%]) *****
1560     '
1570     HEAT%=YEXP*0.5+100      :     ' --- Basic Value ---
1580     '
1590       if YEXP>=0 and B>=0 then HEAT%=HEAT%*2
1600       if B<=-0.1 then HEAT%=HEAT%/3 else goto 1610
1610       if B>=0.1 then HEAT%=HEAT%+A else if B>=0 then HEAT%=HEAT%+0.5*A
1620     '
1630     ' --- Comparison of [B] to [BB] ---
1640       if BB>=B then goto 1680 else print "*";
1650       if BB<0 then goto 1660 else goto 1670
1660       if B<0 then HEAT%=HEAT%+YEXP*0.5 else HEAT%=HEAT%+YEXP*0.8
           :  goto 1700
1670       HEAT%=HEAT%+YEXP*1.2    :    goto 1700
1680     '
1690      if BB<=0 and YEXP<100 then HEAT%=HEAT%*0.8 else HEAT%=HEAT%
1700     BB=B
1710     '
1720     ' --- Limitation of [HEAT%], 0 to 1000 ---
1730        if HEAT%<0 then HEAT%=0
1740        if HEAT%>1000 then HEAT%=1000
1750     '
1760 return
```